



第7章 マイコンに効率良く 仕事をさせるしくみ

割り込み処理の基本を マスタしよう!

山本 秀樹
Hideki Yamamoto

本章では**割り込み**について取り上げます。その前に図1のようなたとえ話で割り込みを考えてみます。

仕事に、重要な要件の電話がかかってくるのを待っているとします。もしその電話機は着信音が鳴らず、着信時に光るだけだとすると、(a)のように仕事に目で確認する必要があります。しかし着信音が鳴るようになっていれば、いちいち確認する必要がなく、(b)のように音が鳴るまで仕事に専念できます。

第5章では、スイッチを押すとLEDが点灯するプログラムを作りました。そのプログラムは、スイッチの状態をループ中でつねに確認していることから、図1で言えば(a)のパターンに相当します。これは、**ポーリング**とも呼ばれます。

割り込みを使うと、何か別の処理をずっと行っている、スイッチが押されるとそれが通知されるので、(b)のパターンのプログラムを作成できます。

プログラムを作りながら、割り込みのしくみについて見ていきます。

割り込みを使わないプログラム

割り込みを使う場合と比較するために、まずリスト1の割り込みを使わないプログラムを作ります。このプログラムは、**2sごとにLED₁の点灯・消灯を繰り返し、SW₂が押されたことを一度でも検出するとLED₂を点灯**します。

SW₂が押されているかどうかは、LED₁を点灯する直前に判定しています。したがって、それ以外のタイミングでSW₂を押しても、LED₂は点灯しません。

このプログラムを実行したときは、一度LED₂が点灯すると元に戻す手段がないので、**再実行するにはHEWからリセットしてやり直してください。くれぐれも、実験基板のリセット・スイッチは使わないでください。**

プログラムの中の新規要素としては、SW₂が押されたことがあるかどうかの処理です。**スイッチが押さ**



(a) 着信で光る電話の場合(割り込みなしのイメージ)

(b) 着信で音が出る電話の場合(割り込みありのイメージ)

図1 割り込みの有無のイメージ

Keywords

割り込み、CMP命令、BMNZ命令、BMZ命令、BMCnd命令、Zフラグ、Iフラグ、スタック・ポインタ、ISP、可変ベクタ・テーブル、固定ベクタ・テーブル、キー入力割り込み、割り込み許可フラグ、割り込みベクタ、レジスタ待避、レジスタ・バンク、割り込みマスク、INT1割り込み、多重割り込み、ノンマスクابل割り込み、ソフトウェア割り込み

リスト1 割り込みを使わない(ポーリングを使った)プログラム

```

; トランジスタ技術 2005年4月号
; 第7章 割り込みを使わない場合

        .INCLUDE     sfr_r815.inc           ; ハードウェア定義ファイルの読み込み

; プログラム部分
        .SECTION    PROGRAM, CODE
        .ORG        0D000h

Start:
        LDC         #0500h, isp             ; (B) ここから実行開始

        JSR         SetClock20MHz          ; 外部クロックに切り替えるサブルーチンを呼び出す
        JSR         InitPort               ; ポートの初期化を行うサブルーチンを呼び出す
        MOV.B       #0, LED20n             ; スイッチ押下フラグをクリアする

Loop:
        BTST        p1_0                   ; スイッチ押下チェック
        JC          SwOff                  ; スイッチが押されていない場合に分岐
        MOV.B       #1, LED20n             ; スイッチ押下フラグをセットする

SwOff:
        CMP.B       #1, LED20n             ; スイッチが押下されたことがあるか
        BMNZ        p1_2                   ; LED 2 に出力する

        BCLR        p1_1                   ; LED 1 を点灯
        JSR         Wait2S                  ; 2秒待ちサブルーチンを呼び出す

        BSET        p1_1                   ; LED 1 を消灯
        JSR         Wait2S                  ; 2秒待ちサブルーチンを呼び出す

        JMP         Loop                    ; LED点滅を繰り返す

; 外部クロックに切り替えるサブルーチン
SetClock20MHz:
        BSET        prc0
        BSET        cm13
        BSET        cm15
        BCLR        cm05
        BCLR        cm16
        BCLR        cm17
        BCLR        cm06
        NOP
        NOP
        NOP
        NOP
        BCLR        ocd2
        BCLR        prc0
        RTS                                ; サブルーチンから復帰する

; ポートの初期化を行うサブルーチン
InitPort:
        MOV.B       #00000110b, drr        ; 駆動能力の設定
        MOV.B       #00000110b, p1         ; ポートに出力する初期値の設定
        MOV.B       #00000110b, pd1        ; ポートの方向を出力に設定
        RTS                                ; サブルーチンから復帰する

; 2秒待ちサブルーチン
Wait2S:
        MOV.W       #200, r1
Wait2S_1:
        JSR         Wait10mS
        SBJNZ.W     #1, r1, Wait2S_1
        RTS

; 10m秒待ちサブルーチン
Wait10mS:
        PUSH.W      r0                      ; R0レジスタを待避する
        MOV.W       #28571, r0
Wait10mS_1:
        SBJNZ.W     #1, r0, Wait10mS_1
        POP.W       r0                      ; R0レジスタをサブルーチン呼び出し前の値に復帰する
        RTS                                ; サブルーチンから復帰する

```

ここでスイッチが押されたかどうかをチェック